

ĐA THỨC

I. CÁC PHÉP TOÁN TRÊN ĐA THỨC.

Với các đa thức ta có thể thực hiện các phép toán như ‘cộng’, ‘trừ’, ‘nhân’, ‘chia’ các đa thức.

Ví dụ : Cho hai đa thức $f(x) = x^2 - 3x + 2$; $g(x) = 4x^3 + x^2 - 3x - 2$.

> **f:=x^2-3*x+2;g:=4*x^3+x^2-3*x-2;**

$$f := x^2 - 3x + 2$$

$$g := 4x^3 + x^2 - 3x - 2$$

Cộng hai đa thức trên ta được:

> **'f+g'=f+g;**

$$f + g = 2x^2 - 6x + 4x^3$$

Trừ đa thức f cho đa thức g ta được:

> **'f-g'=f-g;**

$$f - g = 4 - 4x^3$$

Nhân hai đa thức trên ta được:

> **'f.g'=f*g;**

$$f \cdot g = (x^2 - 3x + 2)(4x^3 + x^2 - 3x - 2)$$

Để xem kết quả khai triển ta dùng hàm > expand(f.g);

> **'f.g'=expand(f*g);**

$$f \cdot g = 4x^5 - 11x^4 + 2x^3 + 9x^2 - 4$$

Chia đa thức f cho đa thức g ta được:

> **'f/g'=f/g;**

$$\frac{f}{g} = \frac{x^2 - 3x + 2}{4x^3 + x^2 - 3x - 2}$$

Dễ nhận thấy f và g có chung nghiệm $x = 1$. Bây giờ để tối giản phân thức $\frac{f}{g}$ ta dùng lệnh

> normal(f/g);

> **'f/g'=normal(f/g);**

$$\frac{f}{g} = \frac{x - 2}{4x^2 + 5x + 2}$$

II. CÁC HÀM LIÊN QUAN ĐẾN ĐA THỨC.

1. Sắp xếp lại một đa thức, danh sách.

Cú pháp: > sort(L)

> sort(L, F)

> sort(A)

> sort(A, V, opt1, opt2, ...)

Trong đó: - L : là một danh sách các giá trị cần sắp xếp.

- A : là một biểu thức đại số.

Ở đây, tôi chỉ giới thiệu việc sắp xếp đa thức.

Ví dụ: Xét đa thức $f(x) = x - 2x^2 + 5x^3 - 7 + 4x^4$.

Ta sắp xếp đa thức trên như sau:

> restart; f:=x-2*x^2+5*x^3-7+4*x^4;

$$f := x - 2x^2 + 5x^3 - 7 + 4x^4$$

> f:=sort(f,x);

$$f := 4x^4 + 5x^3 - 2x^2 + x - 7$$

Để sắp xếp f theo chiều tăng dần (giảm dần) của bậc ta khai báo thêm argument “ascending” (“descending”).

> f:=sort(f,x,ascending);

$$f := -7 + x - 2x^2 + 5x^3 + 4x^4$$

Ví dụ: Xét đa thức $p(x) = y^3 + x^2y^2 + x^3$.

+Sắp xếp đa thức trên theo chiều giảm dần bậc của biến x , ta được:

> restart;p := y^3+y^2*x^2+x^3;
 sort(p,x,descending);

$$x^3 + y^2x^2 + y^3$$

+Sắp xếp đa thức trên theo chiều tăng dần bậc của biến y , ta được:

> sort(p,y,ascending);

$$x^3 + x^2y^2 + y^3$$

+Sắp xếp đa thức trên theo chiều tăng dần bậc của đa thức, ta được:

> sort(p,[x,y],ascending);

$$y^3 + x^3 + x^2y^2$$

+Sắp xếp đa thức trên theo chiều giảm dần bậc của đa thức, ta được:

> sort(p,[x,y],descending);

$$x^2y^2 + x^3 + y^3$$

Ví dụ: Cho đa thức $g(x) = x + 3xy - 2yz + x^2 - 3$.

+Sắp xếp đa thức trên theo thứ tự biến x, y, z và bậc giảm dần, ta được:

> sort(g,[x,y,z],descending);

$$x^2 - 2xy + 3yz + x - 3$$

+Sắp xếp đa thức trên theo thứ tự biến x, z và bậc tăng dần, ta được:

> sort(g,[x,z],ascending);

$$-3 + 3yz - 2yx + x + x^2$$

2. Nhóm các hệ số của các lũy thừa cùng bậc của một đa thức.

Cú pháp: **> collect(a, x, form, func) ;**

Trong đó:

- a : là một đa thức (biểu thức);
- x: là biến hoặc tập hợp các biến hoặc một hàm;
- func: là thủ tục (thường là *simplify* hoặc *factor*);
- form: là tên (thường là *recursive* (đệ quy) hoặc *distributed* (phân phối))

Ví dụ: Đơn giản biểu thức sau bằng cách nhóm các số hạng

$$p(x) = x^2 + 3mx - 5 + mx^2 - x$$

+Đơn giản biểu thức trên bằng cách nhóm các số hạng theo lũy thừa của x:

> p:=x^2+3*m*x-5+m*x^2-x;

$$p := x^2 + 3 m x - 5 + m x^2 - x$$

> p:=collect(p,x);

$$p := -5 + (m + 1) x^2 + (-1 + 3 m) x$$

+Sắp xếp biểu thức trên theo ẩn số x, ta được:

> p:=sort(p,x);

$$p := (m + 1) x^2 + (-1 + 3 m) x - 5$$

+Đơn giản biểu thức trên bằng cách nhóm các số hạng theo lũy thừa của m:

> p:=collect(p,m);

$$p := (x^2 + 3 x) m + x^2 - x - 5$$

+Ta có thể dùng thêm hàm *factor* để phân tích các hệ số thành tích:

> p:=collect(p,m,factor);

$$p := x (x + 3) m + x^2 - x - 5$$

Ví dụ: Cho biểu thức $p(x) = xy + axy + yx^2 - ayx^2 + x + ax$.

+Sắp xếp (đệ quy) biểu thức trên theo biến x, các hệ số chứa y và được sắp xếp theo biến y:

> restart;p := x*y+a*x*y+y*x^2-a*y*x^2+x+a*x;

$$p := x y + a x y + y x^2 - a y x^2 + x + a x$$

> p1:=collect(p,[x,y],recursive);

$$p1 := (1 - a) y x^2 + ((1 + a) y + 1 + a) x$$

+Sắp xếp (đệ quy) biểu thức trên theo biến y, các hệ số chứa x và được sắp xếp theo biến x:

> p2:=collect(p,[y,x],recursive);

$$p2 := ((1 - a) x^2 + (1 + a) x) y + (1 + a) x$$

3. Phân tích một đa thức thành tích của các biểu thức đơn giản nhất.

Cú pháp: **> factor(a, K) ;**

Trong đó: - a: là một biểu thức (biểu thức hữu tỉ).
- K: là từ khoá real hoặc complex; hoặc một số chứa căn; hoặc RootOf.

Ví dụ: Phân tích đa thức sau thành tích $f(x) = x^3 - 3x^2 + 5x - 3$.

```
> restart; f:=x^3-3*x^2+5*x-3;  
f := x3 - 3 x2 + 5 x - 3  
  
> f1:=factor(f);  
f1 := (x - 1) (x2 - 2 x + 3)
```

Tam thức $x^2 - 2x + 3$ không có nghiệm thực, nhưng có 2 nghiệm phức. Vậy nếu phân tích đa thức f trên trường số phức ta sẽ được kết quả:

```
> f2:=factor(f,complex);  
f2 := (x - 1.0000000000 + 1.414213562 I) (x - 1.0000000000 )  
      (x - 1.0000000000 - 1.414213562 I)
```

Bằng cách tìm nghiệm của tam thức $x^2 - 2x + 3$ ta có:

```
> solve(x^2-2*x+3,{x});  
{x = 1 + √2 I}, {x = 1 - √2 I}
```

+Trên cơ sở đó, ta có thể phân tích f theo $\sqrt{2}$ và số phức i :

```
> f3:=factor(f,{sqrt(2),I});  
f3 := (x - 1 + √2 I) (x - 1 - √2 I) (x - 1)
```

Ta chú ý có sự khác biệt khi ta nhập $f(x) = x^3 - 3x^2 + 5x - 3.0$, kết quả phân tích sẽ là:

```
> f:=x^3-3*x^2+5*x-3.0;  
f := x3 - 3 x2 + 5 x - 3.0  
  
> factor(f);  
(x - 1.0000000000) (x2 - 2.0000000000 x + 2.9999999999)
```

Ví dụ: Xét đa thức $g(x) = x^3 + 5$.

Nếu chỉ dùng lệnh `>factor(g)`; ta được kết quả:

```
> g:=x^3+5;  
g := x3 + 5  
  
> factor(g);  
x3 + 5
```

Nhưng nếu phân tích trên trường số phức (complex) ta được kết quả:

```
> factor(g,complex);  
(x + 1.709975947) (x - 0.8549879733 + 1.480882610 I)  
(x - 0.8549879733 - 1.480882610 I)
```

Nếu nhập $g(x) = x^3 + 5.0$, thì khi dùng lệnh `>factor(g)`; ta sẽ được kết quả khác:

> **g:=x^3+5.0;**

$$g := x^3 + 5.0$$

> **factor(g);**

$$(x + 1.709975947) (x^2 - 1.709975947 x + 2.924017740)$$

Nếu phân tích $g(x) = x^3 + 5$ theo $\sqrt[3]{5}$ hay $5^{1/3}$ ta được kết quả:

> **factor(g, 5^(1/3));**

$$(x^2 - x 5^{(1/3)} + 5^{(2/3)}) (x + 5^{(1/3)})$$

Ví dụ: Xét đa thức $p(x) = x^4 - 2$.

Nếu chỉ dùng lệnh **>factor(g);** ta được kết quả:

> **p:=x^4-2;**

$$p := x^4 - 2$$

> **factor(p);**

$$x^4 - 2$$

Nhưng nếu phân tích trên trường số phức (complex) ta được kết quả:

> **factor(p, complex);**

$$(x + 1.189207115) (x + 1.189207115 I) (x - 1.189207115 I) (x - 1.189207115)$$

Nếu phân tích p theo $\sqrt{2}$, ta được:

> **factor(p, sqrt(2));**

$$(x^2 + \sqrt{2}) (x^2 - \sqrt{2})$$

Nếu phân tích p theo $\sqrt[4]{2}$ (hay $2^{1/4}$) và số phức i , ta được:

> **factor(p, {root(2,4), I});** $\#\{\text{root}(2,4) \text{ là } \sqrt[4]{2}\}$

$$(x + 2^{(1/4)} I) (x - 2^{(1/4)} I) (x - 2^{(1/4)}) (x + 2^{(1/4)})$$

Cũng có thể nhập như sau:

> **factor(p, {2^(1/4), I});**

$$(x + 2^{(1/4)} I) (x - 2^{(1/4)} I) (x - 2^{(1/4)}) (x + 2^{(1/4)})$$

●●● Ngoài hàm **factor** ta còn có thể dùng hàm **split** trong gói lệnh **with(polytools)** để phân tích một biểu thức (đa thức) thành tích các biểu thức đơn giản:

Cú pháp: > **with(polytools):**
 > **split(a,x,b);**

Trong đó: - a: là biểu thức (đa thức);
 - x : là biến.
 - b: là biến được gán cho kết quả thu được.

Ví dụ: Phân tích biểu thức $x^2 + x + 1$ thành tích.

Dùng gói lệnh trên và hàm split, ta có kết quả:

```
> with(polytools);  
> split(x^2+x+1,x);  
      (x - RootOf(_Z^2 + _Z + 1)) (x + 1 + RootOf(_Z^2 + _Z + 1))
```

Để thấy kết quả cụ thể hơn ta làm tiếp:

```
> allvalues(%);  
      { (x + 1/2 - 1/2 I sqrt(3)) (x + 1/2 + 1/2 I sqrt(3)) }, { (x + 1/2 - 1/2 I sqrt(3)) (x + 1/2 + 1/2 I sqrt(3)) }
```

Hoặc:

```
> evalf(%);  
(x + 0.5000000000 - 0.8660254040 I) (x + 0.5000000000 + 0.8660254040 I),  
(x + 0.5000000000 - 0.8660254040 I) (x + 0.5000000000 + 0.8660254040 I)
```

Nhận xét: Với gói lệnh này, kết quả thu được rất chi tiết, đầy đủ hơn so với dùng hàm factor sau khi dùng thêm hàm allvalues(%);

4. Khai triển một đa thức.

Cú pháp: **> expand(expr, expr1, expr2, ..., exprn);**

Trong đó: - expr: là đa biểu thức đại số bất kỳ (dạng tích, lũy thừa, lượng giác,...) muốn khai triển.

Ví dụ: Khai triển đa thức: $p(x) = (x-1)(3x+2) + x^2 - x$

```
> p := (x-1)*(3*x+2)+x^2-x;  
      p := (x - 1) (3 x + 2) + x^2 - x
```

```
> p:=expand(p);  
      p := 4 x^2 - 2 x - 2
```

Phân tích kết quả trên thành tích:

```
> factor(p);  
      2 (2 x + 1) (x - 1)
```

Sau đó khai triển kết quả thu được theo $x-1$, ta được:

```
> expand(%,x-1);  
      4 (x - 1) x + 2 x - 2
```

Ví dụ: Giải phương trình $\sqrt{2x-1} + \sqrt{x+3} = 3$.

Theo phương pháp thông thường, ta giải phương trình này bằng cách bình phương hai vế sau khi đã tìm tập xác định (điều kiện xác định) cho phương trình.

Điều kiện: $\begin{cases} 2x-1 \geq 0 \\ x+3 \geq 0 \end{cases} \Leftrightarrow \begin{cases} x \geq 1/2 \\ x \geq -3 \end{cases} \Leftrightarrow x \geq \frac{1}{2}.$

Bây giờ nhập phương trình vào Maple cùng các bước giải phương trình trên như sau:

```
> restart;eq:=sqrt(2*x-1)+sqrt(x+3)=3:eq;  
      
$$\sqrt{2x-1} + \sqrt{x+3} = 3$$
  
  
> `Bình phương hai ve cua  
PT:`;a:=(lhs(eq))^2;b:=(rhs(eq))^2;a=b;  
      Bình phương hai ve cua PT:  
      
$$(\sqrt{2x-1} + \sqrt{x+3})^2 = 9$$
  
  
> `Khai trien ta duoc:`;a:=expand(a):a=b;  
      Khai trien ta duoc:  
      
$$3x + 2 + 2\sqrt{2x-1}\sqrt{x+3} = 9$$
  
  
> `PT tương đương voi:`;c:=a-(op(1,a)+op(2,a)): b:=expand(b-  
(op(1,a)+op(2,a))):c=b;  
      PT tương đương voi:  
      
$$2\sqrt{2x-1}\sqrt{x+3} = 7 - 3x$$
  
  
> `Bình phương hai ve ta duoc:`;c:=c^2;b:=b^2:c=b;  
      Bình phương hai ve ta duoc:  
      
$$4(2x-1)(x+3) = (7-3x)^2$$
  
  
> `Khai trien va rut gon, ta duoc PT tương  
duong:`;eq:=sort(expand(c-b),x):eq=0;  
      Khai trien va rut gon, ta duoc PT tương đương:  
      
$$-x^2 + 62x - 61 = 0$$
  
  
> `Tập nghiệm của PT này là:`;T:={solve(eq, {x})};  
      Tập nghiệm của PT này là:  
      
$$T := \{ \{x = 1\}, \{x = 61\} \}$$

```

Với đoạn lệnh trên, ta có thể giải các phương trình có dạng tương tự bằng cách nhập lại phương trình trong dòng lệnh đầu tiên (khai báo eq:=).

Quý bạn đọc có thể giải các phương trình :

1) $\sqrt{x+3} + 1 = \sqrt{3x-1}$

2) $\sqrt{x-1} - \sqrt{5x-1} = \sqrt{3x-2}$

5. Rút gọn hệ số, trích hệ số rút gọn của một đa thức.

a) Rút gọn hệ số của đa thức poly:

Cú pháp: `> primpact(poly,x,'co');`

Trong đó: - poly: là đa thức

- x : là biến hoặc tập hợp các biến.
- co: là tên của hệ số cần làm gọn.

Ví dụ 1: Rút gọn hệ số của đa thức $p(x) = \frac{x^4}{5} - 3x^3 + x^2 - \frac{3x}{2} + 1$ ta được:

```
> restart;p:=x^4/5-3*x^3+x^2-3*x/2+1;
```

$$p := \frac{1}{5}x^4 - 3x^3 + x^2 - \frac{3}{2}x + 1$$

```
> primpart(p,x,'co'):`Da thuc rut gon`=%;
```

$$Da thuc rut gon = 2x^4 - 30x^3 + 10x^2 - 15x + 10$$

Nếu muốn biết “hệ số đã rút gọn” ta khai báo argumen ‘co’ trong câu lệnh:

```
> primpart(p,x,'co'):`Da thuc rut gon`=%;
```

$$Da thuc rut gon = 2x^4 - 30x^3 + 10x^2 - 15x + 10$$

```
> `He so rut gon`=co;
```

$$He so rut gon = \frac{1}{10}$$

Ví dụ 2: Cho đa thức (nhiều biến) $p(x, y) = 3xy + 6x^2y - 12y^2$.

Làm gọn đa thức trên theo biến x, ta được:

```
> restart;p:=3*x*y+6*x^2*y-12*y^2;
```

$$p := 3xy + 6x^2y - 12y^2$$

```
>> primpart(p,x,'co'):`Da thuc rut gon`=%;`He so rut gon`=co;
```

$$Da thuc rut gon = -4y + x + 2x^2$$

$$He so rut gon = 3y$$

Làm gọn đa thức trên theo biến y, ta được:

```
> restart;p:=3*x*y+6*x^2*y-12*y^2;
```

$$p := 3xy + 6x^2y - 12y^2$$

```
> primpart(p,y,'co'):`Da thuc rut gon`=%;`He so rut gon`=co;
```

$$Da thuc rut gon = xy + 2x^2y - 4y^2$$

$$He so rut gon = 3$$

Làm gọn đa thức trên theo biến x và y, ta được:

```
> restart;p:=3*x*y+6*x^2*y-12*y^2;
```

$$p := 3xy + 6x^2y - 12y^2$$

```
>> primpart(p,[x,y],'co'):`Da thuc rut gon`=%;`He so rut gon`=co;
```

$$Da thuc rut gon = xy + 2x^2y - 4y^2$$

He so rut gon = 3

b) Trích hệ số rút gọn:

Cú pháp: **> content(poly,x,'pp');**

Trong đó: - poly: là đa thức
- x : là biến hoặc tập hợp các biến.
- pp: là tên của đa thức thu được sau khi trích hệ số rút gọn.

Ví dụ: Cho đa thức $p = 3xy - 6x$.

Trích hệ số rút gọn của đa thức trên theo biến x và tìm đa thức thu được sau khi rút gọn:

> restart;p:=3*x*y-2*x;

p := 3 x y - 2 x

> content(p,x,'pp'):`He so rut gon`=%;`Da thuc thu duoc`=pp;

He so rut gon = 3 y - 2

Da thuc thu duoc = x

Trích hệ số rút gọn của đa thức trên theo biến y và tìm đa thức thu được sau khi rút gọn:

> content(p,y,'pp'):`He so rut gon`=%;`Da thuc thu duoc`=pp;

He so rut gon = x

Da thuc thu duoc = 3 y - 2

6.Xác định bậc của một đa thức,biểu thức.

Cú pháp: **> degree(a,x);** _xác định bậc cao nhất của đa thức a.

> ldegree(a,x); _xác định bậc thấp nhất của đa thức a.

Trong đó: - a: là một đa thức;
- x: là biến hoặc tập hợp các biến.

Ví dụ: Xác định bậc của đa thức $p(x) = (x^2 + 1)(3x^3 - x^2 + 2)(2x + 1)$

> p:=(x^2+1)*(3*x^3-3*x^2+2)*(2*x+1);

p := (x^2 + 1) (3 x^3 - 3 x^2 + 2) (2 x + 1)

> `Bac cua da thuc p:`=degree(p,x);

Bac cua da thuc p: = 6

Ví dụ: Xác định bậc cao nhất và thấp nhất của biểu thức: $p(x) = \frac{1}{x^3} - 3x^2 + x^7 - 9$.

> restart;

> p:=1/x^3-3*x^2+x^7-9;

p := \frac{1}{x^3} - 3 x^2 + x^7 - 9

```
> `Bac cao nhat cua bieu thuc p`=degree(p,x);
`Bac thap nhat cua bieu thuc p`=ldegree(p,x);
      Bac cao nhat cua bieu thuc p = 7
```

Bac thap nhat cua bieu thuc p = -3

Với đa thức nhiều biến ta dùng cú pháp:

```
> degree(p(x,y,z,...),{x,y,z,...});
```

Chú ý: Trong Maple 9.5 ta phải khai báo $\{x,y,z,\dots\}$ cho tập hợp các biến chứ không phải $[x,y,z,\dots]$! Nhưng trong Maple 10, Maple 11 thì cả 2 cách khai báo trên đều được.

Ví dụ: Xác định bậc của đa thức: $p(x,y) = 3xy^2 - x^3y + 4x^2 + y^3 + 12$

```
> restart;p:=3*x*y^2-x^3*y+4*x^2+y^2+12;
      p := 3 x y^2 - x^3 y + 4 x^2 + y^2 + 12
```

```
> `Bac cao nhat theo bien x`=degree(p,x);
```

```
`Bac cao nhat theo bien y`=degree(p,y);
      Bac cao nhat theo bien x = 3
```

Bac cao nhat theo bien y = 2

```
> `Bac cao nhat theo bien x va y`=degree(p,{x,y});
```

Bac cao nhat theo bien x va y = 4

```
> `Bac thap nhat theo bien x va y`=ldegree(p,{x,y});
```

Bac thap nhat theo bien x va y = 0

Ví dụ: Cho biểu thức $p(x,y) = \frac{3}{xy^2} + 2xy - 5x^2y$.

```
> restart;p:=3/(x*y^2)-2*x*y-5*x^2*y;
      p := 3 / (x y^2) - 2 x y - 5 x^2 y
```

```
> `Bac cao nhat theo bien x`=degree(p,x);
```

```
`Bac thap nhat theo bien x`=ldegree(p,x);
```

```
`Bac cao nhat theo bien y`=degree(p,y);
```

```
`Bac thap nhat theo bien y`=ldegree(p,y);
```

Bac cao nhat theo bien x = 2

Bac thap nhat theo bien x = -1

Bac cao nhat theo bien y = 1

Bac thap nhat theo bien y = -2

```
> `Bac cao nhat theo bien x va y`=degree(p,{x,y});
```

Bac cao nhat theo bien x va y = 3

```
> `Bac thap nhat theo bien x va y`=ldegree(p,{x,y});  
Bac thap nhat theo bien x va y = -3
```

7.Trích hệ số của một đa thức .

Cú pháp: > **coeff(p,x);**
 > **coeff(p,x,n);**
 > **coeff(p,x^n);**

Trong đó: - p: là đa thức một biến;
 - x: là biến;
 - n: là bậc của lũy thừa của biến x.

Trường hợp muốn trích hệ số tự do (hệ số của x^0) ta dùng lệnh > **coeff(p,x,0);** _không dùng **coeff(p,x^0);**.

Ví dụ: Cho đa thức $f(x) = 3x^2(y+1) - (x+1)(3x-2) + (y^2-1)y$

```
> restart;f:=3*x^2*(y+1)-(x+1)*(3*x-2)+(y^2-1)*y;  
f:=3 x^2 (y+1) - (x+1) (3 x-2) + (y^2-1) y
```

```
> `He so cua x^2 trong f`=coeff(f,x,2);  
He so cua x^2 trong f = 3 y
```

```
> `He so tu do theo bien x`=coeff(f,x,0);  
He so tu do theo bien x = 2 + (y^2-1) y
```

```
> `He so cua y trong f`=coeff(f,y,1);  
He so cua y trong f = 3 x^2 - 1
```

```
> `He so tu do theo bien y`=coeff(f,y,0);factor(%);  
He so tu do theo bien y = 3 x^2 - (x+1) (3 x-2)  
He so tu do theo bien y = -x + 2
```

Chú ý: Tuy nhiên việc trích hệ số của tích x^2y không thực hiện được.

8.Liệt kê các số hạng của một đa thức, biểu thức, danh sách,...

Cú pháp: > **op(f);** - liệt kê các số hạng
 > **nops(f);** - đếm tổng các số hạng.

Trong đó: - f: là một danh sách, biểu thức,...

Ví dụ: Xét đa thức $f(x) = x^3 - 3x + 2$.

•Các số hạng của đa thức là:

```
> f:=x^3-3*x+2;  
`Cac so hang cau thanh f: `[op(f)];  
f:=x^3-3 x+2
```

Các số hạng cấu thành f : $= [x^3, -3x, 2]$

> **`So hang thu nhât:=op(1,f);**

`So hang thu hai:=op(2,f);

So hang thu nhât: $= x^3$

So hang thu hai: $= -3x$

• Tổng số các số hạng của đa thức:

> **`So so hang cua f:=nops(f);**

So so hang cua f : $= 3$

Ví dụ: Xét biểu thức $f(x) = x^2(2x-1)(y+2)$.

Các số hạng cấu thành f là:

> **restart;f:=x^2*(2*x-1)*(y+1);**

`Cac so hang cau thanh f:=[op(f)];

$f := x^2(2x-1)(y+1)$

Các số hạng cấu thành f : $= [x^2, 2x-1, y+1]$

Ví dụ: Cho danh sách: $I = [a, b, c, d]$.

Các phần tử cấu thành I là:

> **L:=[a,b,c,d];**

`Cac phan tu cau thanh L:=op(L);

$L := [a, b, c, d]$

Các phần tử cấu thành L $= (a, b, c, d)$

> **`Tong so cac phan tu:=nops(L);**

Tong so cac phan tu: $= 4$

9. Hàm đồng nhất hệ số tương ứng của hai đa thức, biểu thức.

Cú pháp: > **match(expr=pattern,var,'s');**

- liệt kê các

số hạng

Trong đó: - expr: là đa thức, biểu thức;

- pattern: là mẫu (biểu thức chứa tham số) cần đồng nhất hệ số với các hệ số tương ứng đồng bậc của expr;

- var: là tên của biến trong expr và pattern;

- `s`: là kết quả thu được nếu hàm đồng nhất cho kết quả true.

Ví dụ: Tìm các số a, b, c sao cho hai đa thức sau là bằng nhau:

$f(x) = x^3 - 3x + 2; g(x) = (ax-1)(x^2 + bx + c)$.

> **f:=x^3-3*x+2;g:=(a*x-1)*(x^2+b*x+c);**

$f := x^3 - 3x + 2$

$g := (ax-1)(x^2 + bx + c)$

> **match(f=g,x,'s');**

true

> **s;**

$\{a = 1, c = -2, b = 1\}$

Ví dụ: Khi tính nguyên hàm $I = \int \frac{3\sin x - \cos x}{\sin x + 2\cos x} dx$.

Ta cần biến đổi tử thức $f(x) = 3\sin x - \cos x$ về dạng $f(x) = A.g(x) + B.g'(x)$ hay $f(x) = A(\sin x + 2\cos x) + B(\cos x - 2\sin x)$ với $g(x) = \sin x + 2\cos x$.

Và ta phải tìm các hệ số A, B .

Ta làm như sau:

> **restart;f:=3*sin(x)-**

cos(x);g:=A*(sin(x)+2*cos(x))+B*(cos(x)-2*sin(x));

f:=3 sin(x) - cos(x)

g:=A (sin(x) + 2 cos(x)) + B (cos(x) - 2 sin(x))

> **match(f=g,x,'s');**

true

> **s;**

$\{A = \frac{1}{5}, B = \frac{-7}{5}\}$

Suy ra: $f(x) = \frac{1}{5}[(\sin x + 2\cos x) - 7(\cos x - 2\sin x)]$

Vậy: $I = \frac{1}{5} \int \left(1 - \frac{7(\cos x - 2\sin x)}{\sin x + 2\cos x}\right) dx = \frac{1}{5} \left[x - 7 \int \frac{d(\sin x + 2\cos x)}{\sin x + 2\cos x} \right] + C$

$I = \frac{1}{5} (x - 7 \ln |\sin x + 2\cos x|) + C$

10. Hàm trích các vế (trái/phải) của một phương trình có dạng: $f(x) = g(x)$.

Đặt tên cho phương trình trên là “eq”.

Khi đó: - vế trái phương trình được gọi bằng hàm:

> **lhs(eq);**

- vế phải phương trình được gọi bằng hàm:

> **rhs(eq);**

Ví dụ: Cho phương trình $\sqrt{x-1} + \sqrt{2-x} = 5$

> **eq:=sqrt(x-1)+sqrt(2-x)=5:eq;**

$\sqrt{x-1} + \sqrt{2-x} = 5$

Gọi vế trái, vế phải của phương trình như sau:

> **`Ve trai`=lhs(eq);**

$Ve\ trai = \sqrt{x-1} + \sqrt{2-x}$

> ``Ve phai`=rhs(eq);`

Ve phai = 5

Nhận xét: Ứng dụng Hàm này khi muốn thao tác, biến đổi (khai căn, bình phương,...) trên các vế của một phương trình.

Ví dụ: Giải phương trình $\sqrt{2x^2 + x - 2} = 2 - x$

> `restart;eq:=sqrt(2*x^2+x-2)=2-x:eq;`

$\sqrt{2x^2 + x - 2} = 2 - x$

> `dk:=[solve(rhs(eq)>=0,x)]:`

``Dieu kien`=dk;`

Dieu kien = [RealRange(-∞, 2)]

> `l:=(lhs(eq))^2: r:=(rhs(eq))^2:`

``Binh phuong hai ve duoc:`;`

`eq1:=l=r: eq1;`

Binh phuong hai ve duoc:

$$2x^2 + x - 2 = (2 - x)^2$$

> ``Giai Pt nay thu duoc tap nghiem:`;`

`T:=solve(eq1, {x});`

Giai Pt nay thu duoc tap nghiem:

$$T := \{x = 1\}, \{x = -6\}$$

11. Hàm tính giá trị của một biểu thức (một hoặc nhiều biến).

Cú pháp: > `eval(f,x=a,y=b,...);`

Trong đó: - f: là biểu thức, đa thức;

- x=a; y=b;...: giá trị các biến.

Ví dụ: Cho biểu thức $f(x, y, z) = 3xy + 2xy^2 - 2x^2z$.

> `f:=3*x*y+2*x*y^2-2*x^2*z;`

$$f := 3xy + 2xy^2 - 2x^2z$$

> ``f(2,-1,z)`=eval(f,[x=2,y=-1]);`

$$f(2, -1, z) = -2 - 8z$$

> ``f(2a,-2,3)`=eval(f,[x=2*a,y=-2,z=3]);`

$$f(2a, -2, 3) = 4a - 24a^2$$

12. Phép chia đa thức.

a) **Kiểm tra xem đa thức a có chia hết cho đa thức b hay không.**

Cú pháp: > `divide(a,b,'q');`

Trong đó: - a, b: là các đa thức với hệ số hữu tỉ;

- q: là thương số của phép chia a cho b, nếu kết quả của hàm trên là **true**.

Ví dụ: Kiểm tra xem đa thức $x^4 - y^4$ có chia hết cho đa thức $x + y$ không? Nếu chia hết ta tìm thương số của phép chia đó.

```
> divide(x^4-y^4,x+y,'q');
```

true

```
> `Thuong cua phep chia`;q;
```

```
q:=factor(q):`Hay`;q;
```

Thuong cua phep chia

$$-y^3 + x y^2 - x^2 y + x^3$$

Hay

$$-(-x + y)(y^2 + x^2)$$

• Nếu chỉ muốn kiểm tra xem **a** có chia hết cho **b** hay không thì ta dùng lệnh:

```
> divide(a,b);
```

b) Tìm thương và dư của phép chia đa thức a cho đa thức b.

Giả sử: $a = b.q + r$

• Tìm thương:

Cú pháp:

```
> quo(a,b,x);
```

```
> quo(a,b,x,'r');
```

Trong đó: - a, b: là các đa thức một biến x;

- 'r': là đa thức dư.

• Tìm dư:

Cú pháp:

```
> rem(a,b,x);
```

```
> rem(a,b,x,'q');
```

Trong đó: - a, b: là các đa thức một biến x;

- 'q': là đa thức thương.

Ví dụ: Cho hai đa thức: $f(x) = x^3 + 2x^2 - 3x + 1$; $g(x) = x^2 + x + 2$.

Tìm thương và đa thức dư khi chia f cho g:

```
> f:=x^3+2*x^2-3*x+1;g:=x^2+x+2;
```

$$f := x^3 + 2x^2 - 3x + 1$$

$$g := x^2 + x + 2$$

```
> `Thuong (cua f chia g):`=quo(f,g,x,'r');
```

```
`Da thuc du:`=r;
```

$$\text{Thuong (cua f chia g): } = x + 1$$

$$\text{Da thuc du: } = -1 - 6x$$

Nếu chỉ muốn tìm thương ta chỉ cần khai báo **>quo(f,g,x);** cho đỡ tốn bộ nhớ.

```
> quo(f,g,x):  
`Thuong (cua f chia g):`=%;  
Thuong (cua f chia g): = x + 1
```

Để tìm đa thức dư của phép chia f cho g , ta dùng lệnh: `> rem(f,g,x);`

```
> rem(f,g,x):  
`Du (cua f chia g):`=%;  
Du (cua f chia g): = -1 - 6 x
```

13. Một số gói lệnh tạo sẵn liên quan đến đa thức.

13.a. Gói lệnh: `> with(PolynomialTools):`

• Hàm : `> Translate(f,x,x0);` ;

Công dụng: tính $f(x+x_0)$, với $f(x)$ là đa thức một ẩn cho trước.

Ví dụ: Cho đa thức $f(x) = x^2 + x - 3$

Thay $x = x - 3$ ta được:

```
> with(PolynomialTools):  
f:=x^2+x-3;f1:=Translate(f,x,-3):`f(x-3)`=f1;  
f:=x^2+x-3  
f(x-3)= 3 - 5 x + x^2
```

Xét một ví dụ áp dụng về phép biến đổi đồ thị
(Chương trình Toán lớp 10_Dại số nâng cao):

Cho hàm số $y = f(x) = x^3 - x + 2$. Hỏi phải tịnh tiến đồ thị hàm số đã cho sang trái/phải và lên/xuống bao nhiêu để được đồ thị hàm số $y = g(x) = x^3 + 6x^2 + 11x + 6$?

Hướng dẫn giải:

+ Đầu tiên ta giả sử đồ thị hàm số $g(x)$ được được biến đổi từ đồ thị hàm số $f(x)$ bằng cách tịnh tiến liên tiếp dọc theo trục Ox một đoạn bằng p và theo trục Oy một đoạn bằng q . Khi đó: $g(x) = f(x+p) + q$. (*)

Nhập vào Maple:

```
> restart;  
> with(PolynomialTools):  
f:=x^3-x+2;  
g:=x^3+6*x^2+11*x+6;`-----`;  
f1:=Translate(f,x,p)+q:`f(x+p)+q`=f1;  
f2:=collect(f1,x,factor):`f(x+p)+q`=f2;  
f:=x^3-x+2  
g:=x^3+6 x^2+11 x+6
```


$$f(x+p)+q = x^3 + 3 p x^2 + \frac{(-p + 3 p^3) x}{p} + p^3 + 2 - p + q$$

$$f(x+p)+q = x^3 + 3 p x^2 + (-1 + 3 p^2) x + 2 - p + q + p^3$$

{ Ở trên ta đã giả sử tịnh tiến sang trái/phải một đoạn bằng p và tịnh tiến lên/xuống một đoạn bằng q}.

Để tìm các số thực p, q ta đồng nhất các hệ số tương ứng ở hai vế trong (*):

> **match(g=f2,x,'s');**

true

> **s;**

{ p = 2, q = -2 }

Vậy, đồ thị hàm số g(x) có được bằng cách tịnh tiến liên tiếp đồ thị hàm số f(x) *sang trái* dọc theo trục Ox 2 đơn vị và *xuống dưới* dọc theo trục Oy 2 đơn vị.

14. Một số hàm liên quan đến phân thức hữu tỉ $\left(f(x) = \frac{p(x)}{q(x)} \right)$.

14.a) Hàm trích tử thức và trích mẫu thức:

Cú pháp: > **numer(f);** - trích tử thức của phân thức f;
 > **denom(f);** - trích mẫu thức của phân thức f;

Ví dụ: Cho phân thức $f(x) = \frac{x^2 - 3x + 2}{2x^3 - x^2 + x - 2}$

+Trích tử thức và tử thức của f ta được:

> **f:=(x^2-3*x+2)/(2*x^3-x^2+x-2);**

$$f := \frac{x^2 - 3x + 2}{2x^3 - x^2 + x - 2}$$

> **`Tu thuc la:`;numer(f);**

Tu thuc la:

$$x^2 - 3x + 2$$

> **`Mau thuc la:`;denom(f);**

Mau thuc la:

$$2x^3 - x^2 + x - 2$$

14.b) Hàm đơn giản phân thức hữu tỉ về dạng chuẩn.

Cú pháp: > **normal(f);**

Ví dụ: Với hàm f ở Ví dụ trên, ta có thể làm gọn như sau:

> **f:=normal(f);**

$$f := \frac{x-2}{2x^2+x+2}$$

Nhận xét, với hàm phân thức thì lệnh **> simplify(f);** cho kết quả gần với lệnh **> normal(f);**. Ta xem:

> simplify(f);

$$\frac{x-2}{2x^2+x+2}$$

14c. Phân tích phân thức hữu tỉ thành tổng của các phân thức đơn giản.

Các cú pháp: **> convert(f, parfrac);**
> convert(f, parfrac, K);
> convert(f, parfrac, x);
> convert(f, parfrac, x, K);

Trong đó: - x là biến;
 - K là một trong các option: **real, complex, true, lũy thừa, sqrfree**,

Ví dụ: Cho phân thức $f(x) = \frac{2x^3 + 5x^2 - x - 6}{x^4 + 3x^3 - 3x^2 - 11x - 6}$.

+ Phân tích f thành tổng các phân thức đơn giản

> convert(f, parfrac);

$$\frac{7}{9(x+1)} + \frac{28}{45(x-2)} + \frac{3}{5(x+3)} + \frac{1}{3(x+1)^2}$$

(Vì f chỉ chứa biến x nên trong câu lệnh ta có thể khai báo biến x cũng có thể không cần)

+ Thêm option “**sqrfree**” trong câu lệnh ta có kết quả:

> convert(f, parfrac, sqrfree);

$$\frac{7}{9(x+1)} + \frac{11x+6}{9(x^2+x-6)} + \frac{1}{3(x+1)^2}$$

+ Nếu khai báo option “**real**” hoặc “**complex**” ta có kết quả:

> convert(f, parfrac, real);

$$\frac{0.7777777783}{x+1.} + \frac{0.5999999997}{x+3.000000000} + \frac{0.6222222221}{x-2.000000000} + \frac{0.3333333329}{(x+1.)^2}$$

Nhận xét: Nếu dùng option “**real**” thì kết quả thu được có “tất cả các hệ số dẫn đầu của các mẫu thức đều được giản ước tối đa và bằng 1”.

Ví dụ: Cho phân thức hữu tỉ $f(x) = \frac{x^3 + 5x^2 + 7x + 3}{x^3 - (m+1)x^2 + (m+2)x - 2}$

+ Ta xem kết quả câu lệnh sau:

> convert(f, parfrac);

Error, (in convert/parfrac) the variable name (for conversion to partial fractions) must be provided

Nhận xét:

Vì f có cả ẩn số x và tham số m đó là theo cách hiểu của chúng ta còn Maple chỉ hiểu f là biểu thức hai biến x và m , do đó ta phải khai báo biến cụ thể để Maple hiểu được là ta cần phân tích theo biến nào:

> **convert(f, parfrac, x);**

$$1 - \frac{16}{(-3+m)(x-1)} + \frac{3mx + x^2 - 2x - 17 - 5m}{(-3+m)(x^2 - mx + 2)}$$

+ Nếu khai báo option “true” hoặc “real” hoặc “sqrfree” thì đều có kết quả:

> **convert(f, parfrac, x, true);**

$$1 + \frac{5 + 6x^2 + 5x + mx^2 - mx}{x^3 - mx^2 + 2x - x^2 + mx - 2}$$

♠ Chú ý: Có thể dùng hàm > **factor(f);** để phân tích f thành nhân tử:

> **factor(f);**

$$\frac{(x+1)^2(x+3)}{(x-1)(x^2 - mx + 2)}$$

Ví dụ: Cho phân thức sau:

> **restart; f := (x^4 + 4*x^2 + 2*x^3 + 6*x + 3) / (x^3 - 2*x^2 + 3*x - 2);**

$$f := \frac{x^4 + 4x^2 + 2x^3 + 6x + 3}{x^3 - 2x^2 + 3x - 2}$$

+ Phân tích bình thường ta được:

> **convert(f, parfrac);**

$$x + 4 + \frac{8}{x-1} + \frac{x+5}{x^2 - x + 2}$$

+ Khai báo thêm option “true” hoặc “sqrfree” ta được kết quả:

> **convert(f, parfrac, x, true);**

$$x + 4 + \frac{11 + 9x^2 - 4x}{x^3 - 2x^2 + 3x - 2}$$

> **convert(f, parfrac, sqrfree);**

$$x + 4 + \frac{11 + 9x^2 - 4x}{x^3 - 2x^2 + 3x - 2}$$

+ Khai báo thêm option “real” ta có kết quả:

> **convert(f, parfrac, real);**

$$x + 4 + \frac{7.999999997}{x-1} + \frac{5.000000000 + 1.000000003x}{x^2 - 1.000000000x + 2.000000001}$$

+ Khai báo thêm option “complex” ta có kết quả:

> **convert(f, parfrac, complex);**

$$x + 4 + \frac{7.999999997}{x - 1.} + \frac{0.50000000010 + 2.078804602I}{x - 0.5000000000 + 1.322875656I}$$

$$+ \frac{0.50000000020 - 2.078804602I}{x - 0.5000000000 - 1.322875656I}$$

Ví dụ: Cho phân thức > **f := (2*x^3 - 3*x - 2*x^2 + 3) / (x^5 - 3*x^2 + x^3 - 3) :**

$$f := \frac{2x^3 - 3x - 2x^2 + 3}{x^5 - 3x^2 + x^3 - 3}$$

+ Phân tích f thành tổng các phân thức đơn giản với các option cụ thể ta được:

> **convert(f, parfrac);**

$$\frac{-1 + 2x}{x^2 + 1} - \frac{x(-3 + 2x)}{x^3 - 3}$$

> **convert(f, parfrac, x, true);**

$$\frac{2x^3 - 3x - 2x^2 + 3}{x^5 - 3x^2 + x^3 - 3}$$

> **convert(f, parfrac, real);**

$$\frac{0.03850028555 - 2.026694609x}{x^2 + 1.442249570x + 2.080083822} + \frac{0.02669460800}{x - 1.442249570}$$

$$+ \frac{-0.9999999999 + 2.0000000001x}{x^2 + 1.0000000000}$$

> **convert(f, parfrac, sqrfree);**

$$\frac{2x^3 - 3x - 2x^2 + 3}{x^5 - 3x^2 + x^3 - 3}$$

> **convert(f, parfrac, complex);**

$$\frac{1.0000000000 + 0.5000000000I}{x - 1.0000000000I} + \frac{0.02669460750 - 0.300000000010^{-9}I}{x - 1.442249570}$$

$$+ \frac{1.0000000001 - 0.50000000001I}{x + 1.0000000000I} - \frac{1.013347304 + 0.6004684774I}{x + 0.7211247852 - 1.249024766I}$$

$$- \frac{1.013347304 - 0.6004684783I}{x + 0.7211247852 + 1.249024766I}$$

+ Xem kết quả đầu tiên, biểu thức $x^3 - 3$ có thể phân tích thành nhân tử $x - 3^{(1/3)}$.
Do đó ta khai báo thêm option “**3^(1/3)**”:

> **convert(f, parfrac, 3^(1/3));**

$$- \frac{(-9 + 3 \cdot 3^{(1/3)} + 2 \cdot 3^{(2/3)}) \cdot 3^{(1/3)}}{(x - 3^{(1/3)}) (9 \cdot 3^{(2/3)} + 9)} + \frac{-1 + 2x}{x^2 + 1} + \frac{3 - 2 \cdot 3^{(1/3)} - 4x - x \cdot 3^{(2/3)}}{3(x^2 + x \cdot 3^{(1/3)} + 3^{(2/3)})}$$

+ Tương tự, biểu thức $x^2 + 1$ có thể phân tích trong trường số phức theo nhân tử chứa số phức I (hay $(-1)^{(1/2)}$). Vậy nếu ta khai báo option “ $(-1)^{(1/2)}$ ” sẽ được:

> **convert(f,parfrac,(-1)^(1/2));**

$$\frac{1 + \frac{1}{2}I}{x - I} - \frac{(2x - 3)x}{x^3 - 3} + \frac{1 - \frac{1}{2}I}{x + I}$$

+Nếu khai báo cả 2 option trên ta sẽ được kết quả:

> **convert(f,parfrac,{(-2)^(1/2),3^(1/3)});**

$$\begin{aligned} & \frac{7(\sqrt{-2} + 1)\sqrt{-2}}{4(2 + \sqrt{-2}3^{(1/3)} - 3^{(2/3)})(3^{(1/3)} + \sqrt{-2})(x + \sqrt{-2})} \\ & + \frac{-70x + 27 - 353^{(1/3)} - 243^{(2/3)} - 9x3^{(2/3)} - 12x3^{(1/3)}}{51(x^2 + x3^{(1/3)} + 3^{(2/3)})} \\ & + \frac{7(\sqrt{-2} - 1)\sqrt{-2}}{4(-2 + \sqrt{-2}3^{(1/3)} + 3^{(2/3)})(-3^{(1/3)} + \sqrt{-2})(x - \sqrt{-2})} \\ & + \frac{(-9 + 33^{(1/3)} + 23^{(2/3)})3^{(1/3)}}{9(-3^{(1/3)} + \sqrt{-2})(3^{(1/3)} + \sqrt{-2})(x - 3^{(1/3)})} \end{aligned}$$

Khám phá Maple 11. Đỗ Cao Long. THPT Nam Đông